Reasoning about Actions

Expressive Possibilities

Ira A. Fulton Schools of Engineering
Arizona State University

Objectives

Objective

Use ASP to express more complex transition systems containing nondeterministic actions, concurrent actions and non-inertial fluents

Nondeterministic Actions

Example: Set-up

When Jack goes to work, he either walks there or drives his car. We view walking and driving as two ways of executing the same action. The effect of that action on Jack's location is deterministic, but its effect on the location of his car is not.

Example: Details

- In the transition system represented by this action description we can find two different edges that start at the same state and are labeled by the same event. For instance, there are two edges that start at
- {Loc(Jack)=Home,Loc(Car)=Home} and have the label
  - `{Go(Home)=f, Go(Work)=t}.`
- One of them leads to
  - `{Loc(Jack)=Work, Loc(Car)=Home}      (walking),`
- and the other to
  - `{Loc(Jack)=Work, Loc(Car)=Work}      (driving).`

Example: Computing – 1 of 2

```
% File: 'going.lp'

boolean(t;f).

% sorts and object declarations
object(jack;car).
location(home;work).

% effect and precondition of go
loc(jack,L,T+1) :- go(L,T).
{loc(car,L,T+1)} :- go(L,T), loc(car,L1,T), loc(jack,L1,T),
                    T=0..m-1.
:- go(L,T), loc(jack,L,T).
```

Example: Computing – 2 of 2

```
% domain independent axioms
% fluents are exogenous initially
1{loc(O,LL,0):location(LL)}1 :- object(O).

% uniqueness and existence of fluent values
:- not 1{loc(O,LL,T)}1, object(O), T=1..m.

% actions are exogenous
{go(L,T)} :- location(L), T=0..m-1.

% fluents are inertial
{loc(O,L,T+1)} :- loc(O,L,T), T=0..m-1.
```

Interaction between Concurrent Actions

Example: Set-up

In a standard example of interacting actions, two agents lift the opposite ends of a table upon which various objects have been placed [Pednault, 1987, Section 3]. If one end of the table has been raised, the objects on the table will fall off. But if both ends are lifted simultaneously, the objects on the table will remain fixed.



Example: Details – 1 of 2



```
% File: 'lifting.lp'

% sort and object declarations
boolean(t;f).
end(leftEnd;rightEnd).
height(low;high).

% state constraints
onTable(f,T) :- level(leftEnd,H,T), level(rightEnd,H1,T), H!=H1.

% effect and precondition of lift
level(E,high,T+1) :- lift(E,T).
:- lift(E,T), level(E,high,T).
```

Example: Details – 2 of 2

```
% domain independent axioms
% fluents are exogenous initially
1{level(E,HH,0): height(HH)}1 :- end(E).
1{onTable(BB,0): boolean(BB)}1.

% uniqueness and existence of fluent values
:- not 1{level(E,HH,T)}1, end(E), T=1..m.
:- not 1{onTable(BB,T)}1, T=1..m.

% actions are exogenous
{lift(E,T)} :- end(E), T=0..m-1.

% fluents are inertial
{level(E,H,T+1)} :- level(E,H,T), T=0..m-1.
{onTable(B,T+1)} :- onTable(B,T), T=0..m-1.
```



Non-Inertial Fluents

Example: Set-up

Consider a pendulum that moves from its leftmost position to the rightmost and back, with each swing taking one unit of time.

Example: Details – 1 of 2

```
% File: pendulum

% sorts and object declaration
boolean(t;f).

% effects of hold
right(T+1) :- hold(T), right(T).
left(T+1) :- hold(T), left(T).

% by default, pendulum changes the position
{left(T+1)} :- right(T), T=0..m-1.
{right(T+1)} :- left(T), T=0..m-1.
```

Example: Details – 2 of 2

```
% fluents are exogenous initially
1{right(0);left(0)}1.

% uniqueness and existence of values for fluents
:- not 1{right(T);left(T)}1, T=1..m.

% exogenous action
{hold(T)} :- T=0..m-1.
```

Wrap-Up